

## DFPlayer Mini s PICAXE procesory

<http://www.picaxe.com/docs/spe035.pdf>

Pomocí modulu nelze přímo nahrávat soubory MP3 na kartu micro SD.

**DŮLEŽITÉ** - Všimněte si, že zásuvka karty microSD v horní části modulu MP3 je zásuvka typu push-to-eject (takže musíte kartu lehce zatlačit dovnitř, aby se vysunula). Pokud jednoduše vytáhnete kartu bez jejího vysunutí, můžete způsobit fyzické poškození zásuvky a / nebo modulu!

Soubory MP3 pro hudbu / řeč je možné zkopírovat na kartu pomocí mobilního telefonu, tabletu nebo počítače. Jsou podporovány karty MicroSD od 1 GB do 32 GB není třeba používat rychlé karty, co jsou potřeba pro nahrávání videá apod. Mohou být použity i starší karty z mobilních telefonů.

Hudební soubory musí mít názvy souborů začínající na 0001.mp3, 0002.mp3, 0003.mp3 atd. Můžeme je uložit do složky s názvem \ MP3 na kartě microSD (z tohoto adresáře pak čteme stopy MP3 příkazem 0x12), nebo můžeme použít pojmenování adresářů čísly 01 – 99 a jména souborů musí mít číslo 001 – 255 (a z těchto adresářů spouštíme MP3 pomocí příkazu 0x0F).


Soubory nesmí být chráněny autorskými zákony (DRM).

Jakmile jsou soubory hudby a řeči pečlivě zkopírovány na kartu micro SD, jemně vložte kartu do zásuvky na modulu MP3.


Před použitím karty microSD je vhodné tuto kartu naformátovat rychlým formátováním, file systém FAT, allocation unit size 32 kB.

Pak vytvoříme adresář například MP3. Nejlepší je pak kopírovat soubory najednou (one at time).

### Propojení modulu s PICAXE procesorem



### Testování modulu bez použití PICAXE




1. Ujistěte se, že do MP3 přehrávače byla vložena karta microSD obsahující soubory MP3 (v adresáři MP3, jména souborů číslem).
2. Připojte napájení GND a (4,5V nebo 5V)
3. Krátkým stisknutím tlačítka (next / vol+) se přesunete na další stopu, která se přehraje.
4. Krátkým stisknutím tlačítka (prev/vol -) se přehraje na předchozí stopa (audionahrávka).
6. Dlouhým stisknutím tlačítka (next / vol+) zvýšíte hlasitost.
7. Dlouhým stiskem pravého tlačítka (prev/vol -) snížíte hlasitost.

Všimněte si, že stejně jako u většiny přehrávačů MP3 fungují tlačítka další / předchozí přehrávaná skladby podle pořadí, jak byly soubory původně uloženy na kartě, nikoliv v abecedním pořadí názvů souborů aktuálně na kartě.

Pokud je přehrávání řízeno sériovými příkazy z čipu PICAXE, soubory jsou volány přímo podle čísla souboru, takže pořadí řazení na kartě microSD není důležité.

### Ovládání modulu z procesoru (například PICAXE)

Nejjednodušší třívodičové připojení




```

Symbol TX = B.4
Symbol RX = C.3
Symbol BUSY_PIN = pinC.2
Symbol BAUD_FREQ = M8
Symbol BAUD = T9600_8
Symbol cmd = b0
Symbol arg = w1 ; b3:b2
Symbol arg.lsb = b2
Symbol arg.msb = b3

High TX ;TxD výstup do H
Pause 2000
cmd = $09 ;vyšli do modulu příkaz pro práci s mikroSD kartou
arg = $0002
Gosub Send ;čekej na inicializaci modulu s mikroSD kartou
Pause 4000
cmd = $12 ;přehraj z adresáře \MP3 zvukový soubor 0001.mp3
arg = 0001
Gosub Send
Loop: Pause 1000
goto Loop
Send:
SetFreq BAUD_FREQ
Pause 10
SerOut TX, BAUD, ( $7E, $FF, $06, cmd, $00, arg.msb, arg.lsb, $EF )
SetFreq MDEFAULT
Return

```

Zde procesor nepozná, kdy bylo dokončeno přehrávání předchozí nahrávky. Toto řeší připojení signálu BUSY, jak je vidět na následujícím obrázku. Pokud je přehrávána stopa, je BUSY = 0. Když se nepřehrává, je BUSY=1. Signál, by se tedy mohl také jmenovat „nepřehrává se“.




```

Symbol TX = B.4
Symbol RX = C.3
Symbol BUSY_PIN = pinC.2
Symbol BAUD_FREQ = M8
Symbol BAUD = T9600_8
Symbol cmd = b0
Symbol arg = w1 ; b3:b2
Symbol arg.lsb = b2
Symbol arg.msb = b3
Symbol varA = w2

High TX ;TxD výstup do H
Pause 2000
cmd = $09 ;vyšli do modulu příkaz pro práci s mikroSD kartou
arg = $0002
Gosub Send ;čekej na inicializaci modulu s mikroSD kartou
Pause 4000
cmd = $06 ;Nastav zesílení na 20 (rozsah 0-30)
arg = 20
Gosub Send
Pause 1000
For varA = 1 To 9
  cmd = $12 ;přehraj z adresáře \MP3 zvukový soubor 0001.mp3
  arg = varA
  Gosub Send
  Pause 1000
  Do While BUSY_PIN = 0
    Pause 100
  Loop
Next
Loop:
goto Loop
Send:
SetFreq BAUD_FREQ
Pause 10
SerOut TX, BAUD, ( $7E, $FF, $06, cmd, $00, arg.msb, arg.lsb, $EF )
SetFreq MDEFAULT
Return

```

Pro plnou oboustrannou komunikaci mezi procesorem a modulem je třeba propojit také zpětnou vazbu mezi modulem a procesorem.



**Po zapnutí napájení je potřeba dát modulu čas 4 sec, než mu začneme posílat příkazy.** Za tuto dobu modul načte adresáře z karty a inicializuje se jeho činnost.

## Shrnutí

Názvy souborů a složek jsou uloženy na kartě microSD v desítkovém formátu ASCII čísel, např.

\ MP3 \ 0001.mp3

\ MP3 \ 0123.mp3

Při použití výchozích složek \ MP3 a \ ADVERT musí názvy souborů začínat čtyřmi číslicemi, např.

\ MP3 \ 0001.mp3

\ ADVERT \ 0123.mp3

**Příkaz 0x12** se používá k přehrávání skladeb z výchozí složky \ MP3

**Příkaz 0x13** se používá k přehrávání stop ze složky \ ADVERT

Při použití číslovaných podsložek namísto \ MP3 musí mít název složky 2 číslice (01 - 99) a název souboru musí začínat třemi číslicemi (001 - 255), např.

\ 01 \ 001.wav

\ 02 \ 222.mp3

**Příkaz 0x0F** se používá k přehrávání skladeb z číslované složky

### Reklamní stopy

"Reklama" může být spuštěna během přehrávání jiné skladby ve formátu MP3. Proto můžete například normální stopu ze složky \ MP3 spustit příkazem 0x12. Zatímco tato skladba stále hraje „reklama“ je zahájena příkazem 0x13. Původní skladba bude dočasně pozastavena, zatímco se bude přehrávat reklama. Jakmile je reklama dokončena, původní stopa se automaticky spustí a pokračuje se v jejím přehrávání. Reklamy musí být uloženy v podadresáři \ ADVERT a jsou řízeny těmito příkazy:

play reklamu příkaz **0x13**

zastav přehrávání reklamy příkaz **0x15**

### Souborový systém FAT

Skladby lze také přehrát pomocí odkazu na soubor v systému souborů FAT na paměťové kartě (tj. Pořadí, ve kterém byly nové soubory uloženy na kartu po formátování). Tento systém souborů FAT zcela ignoruje skutečný název souboru. Platné hodnoty souborů souborů FAT jsou 01-255.

Příkaz **0x03** se používá k přehrávání určitého čísla souboru FAT

Příkazy **0x01 a 0x02** se používají k přehrávání dalších / předchozích čísel souborů FAT

**Tlačítka next/prev** (další skladba, předchozí skladba) také používají čísla souboru FAT a ignorují skutečné názvy souborů.

## Formát příkazu

Přenosová rychlost je 9600, N, 8,1 (nečinnost vysoká, což je nastavení příkazu PICAXE serout T9600\_8)

Obecný formát příkazu je

Start	vv	ln		fb					
<b>7E</b>	<b>FF</b>	<b>06</b>	cm	<b>00</b>	dh	dl	ch	cl	<b>EF</b>

(vv) Verze = FF

(ln) Počet bajtů vv + ln + cm + fb + dh + dl = 06

(fb) Použijte 01, pokud chcete zpětnou odezvu z modulu na příkaz do modulu poslaný (feed back)

cm = povel

dh dl = data pro příkaz

ch cl = kontrolní součet = 0 - vv - ln - cm - fb - dh - dl

**Poznámka:** Některé technické listy obsahují kontrolní součet "ch" a "cl" jako součást sériových dat. Zdá se však, že modul úplně ignoruje kontrolní součet a je jedno, zda je či není použit kontrolní součet. Proto je pro jednoduchost vynechán ve všech příkladech PICAXE, která proto dává následující formát sekvence příkazu

<b>7E</b>	<b>FF</b>	<b>06</b>	<i>cm</i>	<b>00</b>	<i>dh</i>	<i>dl</i>	<b>EF</b>
-----------	-----------	-----------	-----------	-----------	-----------	-----------	-----------

Všimněte si, že čísla názvů souborů v seznamu sériových příkazů jsou uvedena v hexadecimálním kódu (ne desítkovém)

*Příklad:* přehrávání souboru

\ 02 \ 222.mp3

což je složka 02 (0x02) stopa 222 (0xDE) - sekvence příkazů bajtů by byla

*serout pin, T9600\_8, (\$ 7E, \$ FF, \$ 06, \$ 0F, \$ 02, \$ DE, \$ EF)*

Přehled příkazů

7E	FF	06	cm	00	dh	dl	EF
----	----	----	----	----	----	----	----

01	00	00					Přehrajte další číslo souboru FAT
02	00	00					Přehraje předchozí číslo souboru FAT
03	00	nn					Přehrát číslo souboru FN nn = 01-FF
04	00	00					Zvyšte hlasitost
05	00	00					Snížení hlasitosti
06	00	vv					Nastavení hlasitosti, vv = 00h-1Eh (1Eh = maximální zesílení)
07	00	00					EQ Normální
07	00	01					EQ Pop
07	00	02					EQ Rock
07	00	03					Jazz
07	00	04					EQ Classic

07	00 05	EQ Base
08	00 nn	Opakované přehrávání FAT čísla souboru nn = 01-FF
09	00 01	Zdroj je USB klíč
09	00 02	Zdroj je karta TF (microSD)
09	00 03	Zdroj je Aux (nepoužívá se)
09	00 04	Zdroj je PC (režim ladění)
09	00 05	Zdroj je Flash paměť
09	00 06	Zdroj je režim spánku
0A	00 00	Pohotovostní režim (vstup do režimu spánku)
0B	00 00	Normální režim (ukončení režimu spánku)
0C	00 00	Reset modulu
0D	00 00	Pokračování v přehrávání (bez pauzy)
0E	00 00	Pause
0F	ff tt	Přehrávání ze složky ff = 01-63, track tt = 01-FF
10	00 vv	Nastavení zesílení, vv = 00-1F
10	01 vv	Vnější nastavení hlasitosti, vv = 00-1F
11	00 00	Normální režim
11	00 01	Opakované přehrávání (další skladba při přehrávání smyček)
12	tt tt	Přehrávání ze složky MP3, track tt = 0001-270F
13	tt tt	Přehrávání ze složky ADVERT, track tt = 0001-270F
14	ft tt	Přehrát ze složky f = 1-F, track ttt = 001-3E7
15	00 00	Zastavte přehrávání reklamy, obnovte přehrávání původní stopy
16	00 00	Zastavte přehrávání
17	00 ff	Opakované přehrávání skladeb ve složce ff. 0x16 k zastavení
18	00 00	Shuffle přehraje všechny stopy. Příkaz 0x16 zastav přehrávání.
19	00 00	Začněte smyčkou právě přehrávané stopy
19	00 01	Ukončení smyčky aktuálně přehrávané stopy
1A	00 00	Ukončení Mute DAC (Audio) (povolit DAC - Audio)
1A	00 01	Mute DAC (Audio) (vypnout DAC, nastavit DAC high-Z)

kde

vv = volume


ff = folder

tt = track

nn = FAT file number

Na [www.picaxe.com/docs/spe033.pdf](http://www.picaxe.com/docs/spe033.pdf) je kompletní technická dokumentace modulu.

Nyní si uděláme malou úpravu modulu, aby se nám dobře používat s našimi systémy. Uděláme si z něj H&S Modul, ke kterému můžeme připojit přímo reproduktor 8Ω, 3W. Osazovák je na následujícím obrázku.



Výše je popsané ovládání modulu pomocí basicu mikrokontroléru PICAXE.


Jednoduchý příklad v PICAXE BASICU přehraje skladbu poloviční hlasitostí a po jejím přehrání čeká na stisk tlačítka „přehraj skladbu znovu“.

```

Symbol TX = B.4
Symbol BUSY_PIN = pinC.2
Symbol cmd = b0
Symbol arg2 = b2
Symbol arg1 = b1
Symbol tlZnovu = pinC.1
;*****
; Inicializace modulu MP3, bez n? modul s procesorem nefunguje
;.....
SetFreq M8 ;nastavi frekvenci hodin procesoru na 8 MHz
dirsB = $ff ;port B jako vystupy B.0 z B.7
Pause 4000
cmd = $09
arg2 = $00
arg1 = $02
Gosub Send ;vysli prikaz k praci s SD RAM do modulu
Pause 4000
cmd = $06
arg2 = 00
arg1 = 15
Gosub Send ;nastav hlasitost modulu 0-30 (0h-1eh)
pause 1000
;*****
znovu:
cmd = $12 ;prehraj skladbu cislo 0002 z adresare MP3
arg2 = 00
arg1 = 02
Gosub Send
cekej:
if BUSY_PIN = 1 then goto stiskTl
goto cekej
stiskTl:
;dohraje-li skladba, ceká se na stisk
; tlZnovu
;po stisku tlačítka se skladba prehraje
;znovu
if tlZnovu=1 then goto znovu
goto stiskTl
;.....
; Podprogram odesle 8 bytu dat s prikazem do modulu MP3 9600 Bd, 8 bitu, 1 stop, bez parity
;.....
Send:
SerOut TX, T9600_8, ( $7E, $FF, $06, cmd, $00, arg2, arg1, $EF)
Return

```

Na následujícím obrázku je tentýž program, pro změnu napsaný ve flowchart editoru. Tak jak je vidět i úplný začátečník může modul řídit z PICAXE procesoru.




Je možné si zjednoduší budoucí využití podprogramů po danou aplikaci, čímž se zkrátí zápis hlavního programu, využívajícího modul. V předchozím příkladu je při každém spuštění skladby třeba naplnit 3 parametry (cmd, arg2 a arg 1). Totéž platí i pro inicializaci modulu. Pokud pro inicializaci modulu a spuštění přehrávání z adresáře MP3 vytvoříme vhodné podprogramy, pak pro přehrání jedné skladby ušetříme 4 instrukce v hlavním programu a při každém dalším spuštění nové skladby další instrukci. To nám může, pokud spouštíme při běhu programu víc skladeb, ušetřit poměrně velké množství instrukcí a tím i podstatně zkrátit hlavní program.

```

mp3player20M2zjednodus.bas
1 Symbol TxD = B.4
2 Symbol NBSY = pinC.0
3 Symbol arg2 = b2
4 Symbol arg1 = b1
5 ;*****
6 ; Inicializace modulu MP3, bez n? modul s procesorem nefunguje
7 ;*****
8 SetFreq M8 ;nastavi frekvenci hodin procesoru na 8 MHz
9 dirsB = $ff ;port B jako vystupy B.0 z B.7
10 Pause 4000
11 Gosub initMP3mod ;vysli prikaz k praci s SD RAM do modulu
12 Pause 4000
13 arg2 = 00 ;zacni prehravat skladbu cislo 0002 z adresare MP3
14 arg1 = 02
15 Gosub playMP3
16 pause 200
17 cekej:
18 if NBSY = 1 then goto dalsi
19 goto cekej
20 dalsi:
21 arg2 = 00 ;cislo skladby v MP3
22 arg1 = 02
23 Gosub playMP3
24 pause 200
25 cekej1:
26 if NBSY = 1 then goto cekej1
27 goto cekej1
28 ;*****
29 ; Podprogram odesle 8 bytu dat s prikazem do modulu MP3 9600 Bd, 8 bitu, 1 stop, bez parity
30 ;*****
31 initMP3mod:
32 SerOut TxD, T9600_8, ( $7E, $FF, $06, $09, $00, $00, $02, $EF)
33 Return
34 playMP3:
35 SerOut TxD, T9600_8, ( $7E, $FF, $06, $12, $00, arg2, arg1, $EF)
36 Return

```





Podobně, pokud se jména skladeb liší pouze v arg1, není třeba pořád dokola arg2 plnit obsahem a ušetří se zde další instrukce.